

NAG Fortran Library Routine Document

C05PCF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

C05PCF is a comprehensive routine to find a solution of a system of nonlinear equations by a modification of the Powell hybrid method. The user must provide the Jacobian.

2 Specification

```

SUBROUTINE C05PCF(FCN, N, X, FVEC, FJAC, LDFJAC, XTOL, MAXFEV, DIAG,
1          MODE, FACTOR, NPRINT, NFEV, NJEV, R, LR, QTF, W,
2          IFAIL)
    INTEGER          N, LDFJAC, MAXFEV, MODE, NPRINT, NFEV, NJEV, LR, IFAIL
    real            X(N), FVEC(N), FJAC(LDFJAC,N), XTOL, DIAG(N), FACTOR,
1          R(LR), QTF(N), W(N,4)
    EXTERNAL        FCN

```

3 Description

The system of equations is defined as:

$$f_i(x_1, x_2, \dots, x_n) = 0, \quad \text{for } i = 1, 2, \dots, n.$$

C05PCF is based upon the MINPACK routine HYBRJ (Moré *et al.* (1980)). It chooses the correction at each step as a convex combination of the Newton and scaled gradient directions. Under reasonable conditions this guarantees global convergence from starting points far from the solution and a fast rate of convergence. The Jacobian is updated by the rank-1 method of Broyden. At the starting point the Jacobian is calculated, but it is not recalculated until the rank-1 method fails to produce satisfactory progress. For more details see Powell (1970).

4 References

Moré J J, Garbow B S and Hillstom K E (1980) User guide for MINPACK-1 *Technical Report ANL-80-74* Argonne National Laboratory

Powell M J D (1970) A hybrid method for nonlinear algebraic equations *Numerical Methods for Nonlinear Algebraic Equations* (ed P Rabinowitz) Gordon and Breach

5 Parameters

1: FCN – SUBROUTINE, supplied by the user. *External Procedure*

Depending upon the value of IFLAG, FCN must either return the values of the functions f_i at a point x or return the Jacobian at x .

Its specification is:

<pre> SUBROUTINE FCN(N, X, FVEC, FJAC, LDFJAC, IFLAG) INTEGER N, LDFJAC, IFLAG real X(N), FVEC(N), FJAC(LDFJAC,N) </pre>		
1:	N – INTEGER	<i>Input</i>
	<i>On entry:</i> the number of equations, n .	

2:	X(N) – <i>real</i> array	<i>Input</i>
	<i>On entry:</i> the components of the point at which the functions or the Jacobian must be evaluated.	
3:	FVEC(N) – <i>real</i> array	<i>Output</i>
	<i>On exit:</i> if IFLAG = 1 on entry, FVEC must contain the function values $f_i(x)$ (unless IFLAG is set to a negative value by FCN). If IFLAG = 0 or 2 on entry, FVEC must not be changed.	
4:	FJAC(LDFJAC,N) – <i>real</i> array	<i>Output</i>
	<i>On exit:</i> if IFLAG = 2 on entry, FJAC(i, j) must contain the value of $\frac{\partial f_i}{\partial x_j}$ at the point x , for $i, j = 1, 2, \dots, n$ (unless IFLAG is set to a negative value by FCN). If IFLAG = 0 or 1 on entry, FJAC must not be changed.	
5:	LDFJAC – INTEGER	<i>Input</i>
	<i>On entry:</i> the first dimension of FJAC.	
6:	IFLAG – INTEGER	<i>Input/Output</i>
	<i>On entry:</i> IFLAG = 0, 1 or 2: if IFLAG = 0, X and FVEC are available for printing (see NPRINT below); if IFLAG = 1, FVEC is to be updated; if IFLAG = 2, FJAC is to be updated. <i>On exit:</i> in general, IFLAG should not be reset by FCN. If, however, the user wishes to terminate execution (perhaps because some illegal point X has been reached), then IFLAG should be set to a negative integer. This value will be returned through IFAIL.	

FCN must be declared as EXTERNAL in the (sub)program from which C05PCF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

- | | | |
|----|---|---------------------|
| 2: | N – INTEGER | <i>Input</i> |
| | <i>On entry:</i> the number of equations, n .
<i>Constraint:</i> $N > 0$. | |
| 3: | X(N) – <i>real</i> array | <i>Input/Output</i> |
| | <i>On entry:</i> an initial guess at the solution vector.
<i>On exit:</i> the final estimate of the solution vector. | |
| 4: | FVEC(N) – <i>real</i> array | <i>Output</i> |
| | <i>On exit:</i> the function values at the final point, X. | |
| 5: | FJAC(LDFJAC,N) – <i>real</i> array | <i>Output</i> |
| | <i>On exit:</i> the orthogonal matrix Q produced by the QR factorization of the final approximate Jacobian. | |
| 6: | LDFJAC – INTEGER | <i>Input</i> |
| | <i>On entry:</i> the first dimension of the array FJAC as declared in the (sub)program from which C05PCF is called.
<i>Constraint:</i> $LDFJAC \geq N$. | |

- 7: XTOL – *real* Input
On entry: the accuracy in X to which the solution is required.
Suggested value: the square root of the *machine precision*.
Constraint: $XTOL \geq 0.0$.
- 8: MAXFEV – INTEGER Input
On entry: the maximum number of calls to FCN with IFLAG $\neq 0$. C05PCF will exit with IFAIL = 2, if, at the end of an iteration, the number of calls to FCN exceeds MAXFEV.
Suggested value: $MAXFEV = 100 \times (N + 1)$.
Constraint: $MAXFEV > 0$.
- 9: DIAG(N) – *real* array Input/Output
On entry: if MODE = 2 (see below), DIAG must contain multiplicative scale factors for the variables.
Constraint: $DIAG(i) > 0.0$, for $i = 1, 2, \dots, n$.
On exit: the scale factors actually used (computed internally if MODE $\neq 2$).
- 10: MODE – INTEGER Input
On entry: indicates whether or not the user has provided scaling factors in DIAG. If MODE = 2, the scaling must have been specified in DIAG. Otherwise, the variables will be scaled internally.
- 11: FACTOR – *real* Input
On entry: a quantity to be used in determining the initial step bound. In most cases, FACTOR should lie between 0.1 and 100.0. (The step bound is $FACTOR \times \|DIAG \times X\|_2$ if this is non-zero; otherwise the bound is FACTOR.)
Suggested value: FACTOR = 100.0.
Constraint: FACTOR > 0.0.
- 12: NPRINT – INTEGER Input
On entry: indicates whether or not special calls to FCN with IFLAG = 0 are to be made for printing purposes. If NPRINT ≤ 0 , then no calls are made. If NPRINT > 0, then FCN is called at the beginning of the first iteration, every NPRINT iterations thereafter and immediately prior to the return from C05PCF.
- 13: NFEV – INTEGER Output
On exit: the number of calls made to FCN to evaluate the functions.
- 14: NJEV – INTEGER Output
On exit: the number of calls made to FCN to evaluate the Jacobian.
- 15: R(LR) – *real* array Output
On exit: the upper triangular matrix R produced by the QR factorization of the final approximate Jacobian, stored row-wise.
- 16: LR – INTEGER Input
On entry: the dimension of the array R.
Constraint: $LR \geq N \times (N + 1)/2$.

- 17: QTF(N) – *real* array *Output*
On exit: the vector $Q^T f$.
- 18: W(N,4) – *real* array *Workspace*
- 19: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.
On exit: IFAIL = 0 unless the routine detects an error (see Section 6).
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL < 0

This indicates an exit from C05PCF because the user has set IFLAG negative in FCN. The value of IFAIL will be the same as the user's setting of IFLAG.

IFAIL = 1

On entry, $N \leq 0$,
 or $XTOL < 0.0$,
 or $MAXFEV \leq 0$,
 or $FACTOR \leq 0.0$,
 or $LDFJAC < N$,
 or $LR < N \times (N + 1)/2$,
 or $MODE = 2$ and $DIAG(i) \leq 0.0$ for some i , $i = 1, 2, \dots, N$.

IFAIL = 2

There have been MAXFEV evaluations of FCN to evaluate the functions. Consider restarting the calculation from the final point held in X.

IFAIL = 3

No further improvement in the approximate solution X is possible; XTOL is too small.

IFAIL = 4

The iteration is not making good progress, as measured by the improvement from the last 5 Jacobian evaluations.

IFAIL = 5

The iteration is not making good progress, as measured by the improvement from the last 10 iterations.

The values IFAIL = 4 and IFAIL = 5 may indicate that the system does not have a zero, or that the solution is very close to the origin (see Section 7). Otherwise, rerunning C05PCF from a different starting point may avoid the region of difficulty.

7 Accuracy

If \hat{x} is the true solution and D denotes the diagonal matrix whose entries are defined by the array DIAG then C05PCF tries to ensure that

$$\|D \times (x - \hat{x})\|_2 \leq \text{XTOL} \times \|D \times \hat{x}\|_2.$$

If this condition is satisfied with $\text{XTOL} = 10^{-k}$, then the larger components of Dx have k significant decimal digits. There is a danger that the smaller components of Dx may have large relative errors, but the fast rate of convergence of C05PCF usually avoids this possibility.

If XTOL is less than the *machine precision* and the above test is satisfied with the *machine precision* in place of XTOL, then the routine exits with IFAIL = 3.

Note: this convergence test is based purely on relative error, and may not indicate convergence if the solution is very close to the origin.

The test assumes that the functions and the Jacobian are coded consistently and that the functions are reasonably well behaved. If these conditions are not satisfied then C05PCF may incorrectly indicate convergence. The coding of the Jacobian can be checked using C05ZAF. If the Jacobian is coded correctly, then the validity of the answer can be checked by rerunning C05PCF with a tighter tolerance.

8 Further Comments

The time required by C05PCF to solve a given problem depends on n , the behaviour of the functions, the accuracy requested and the starting point. The number of arithmetic operations executed by C05PCF is about $11.5 \times n^2$ to process each evaluation of the functions and about $1.3 \times n^3$ to process each evaluation of the Jacobian. Unless FCN can be evaluated quickly, the timing of C05PCF will be strongly influenced by the time spent in FCN.

Ideally the problem should be scaled so that at the solution the function values are of comparable magnitude.

9 Example

To determine the values x_1, \dots, x_9 which satisfy the tridiagonal equations:

$$\begin{aligned} (3 - 2x_1)x_1 - 2x_2 &= -1, \\ -x_{i-1} + (3 - 2x_i)x_i - 2x_{i+1} &= -1, \quad i = 2, 3, \dots, 8 \\ -x_8 + (3 - 2x_9)x_9 &= -1. \end{aligned}$$

9.1 Program Text

Note: the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      C05PCF Example Program Text
*      Mark 14 Revised.  NAG Copyright 1989.
*      .. Parameters ..
      INTEGER          N, LDFJAC, LR
      PARAMETER        (N=9,LDFJAC=N,LR=(N*(N+1))/2)
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
*      .. Local Scalars ..
      real             FACTOR, FNORM, XTOL
      INTEGER          IFAIL, J, MAXFEV, MODE, NFEV, NJEV, NPRINT
*      .. Local Arrays ..
      real             DIAG(N), FJAC(LDFJAC,N), FVEC(N), QTF(N), R(LR),
+                     W(N,4), X(N)
*      .. External Functions ..
      real             F06EJF, X02AJF
      EXTERNAL          F06EJF, X02AJF
*      .. External Subroutines ..
      EXTERNAL          C05PCF, FCN
*      .. Intrinsic Functions ..
```

```

      INTRINSIC          SQRT
*      .. Executable Statements ..
      WRITE (NOUT,*) 'C05PCF Example Program Results'
      WRITE (NOUT,*)
*      The following starting values provide a rough solution.
      DO 20 J = 1, N
        X(J) = -1.0e0
20    CONTINUE
      XTOL = SQRT(X02AJF())
      DO 40 J = 1, N
        DIAG(J) = 1.0e0
40    CONTINUE
      MAXFEV = 1000
      MODE = 2
      FACTOR = 100.0e0
      NPRINT = 0
      IFAIL = 1
*
      CALL C05PCF(FCN,N,X,FVEC,FJAC,LDFJAC,XTOL,MAXFEV,DIAG,MODE,FACTOR,
+               NPRINT,NFEV,NJEV,R,LR,QTF,W,IFAIL)
*
      IF (IFAIL.EQ.0) THEN
        FNORM = F06EJF(N,FVEC,1)
        WRITE (NOUT,99999) 'Final 2-norm of the residuals =', FNORM
        WRITE (NOUT,*)
        WRITE (NOUT,99998) 'Number of function evaluations =', NFEV
        WRITE (NOUT,*)
        WRITE (NOUT,99998) 'Number of Jacobian evaluations =', NJEV
        WRITE (NOUT,*)
        WRITE (NOUT,*) 'Final approximate solution'
        WRITE (NOUT,*)
        WRITE (NOUT,99997) (X(J),J=1,N)
      ELSE
        WRITE (NOUT,99996) 'IFAIL = ', IFAIL
        IF (IFAIL.GT.2) THEN
          WRITE (NOUT,*)
          WRITE (NOUT,*) 'Approximate solution:'
          WRITE (NOUT,*)
          WRITE (NOUT,99997) (X(J),J=1,N)
        END IF
      END IF
      STOP
*
99999 FORMAT (1X,A,e12.4)
99998 FORMAT (1X,A,I10)
99997 FORMAT (1X,3F12.4)
99996 FORMAT (1X,A,I2)
      END
*
      SUBROUTINE FCN(N,X,FVEC,FJAC,LDFJAC,IFLAG)
*      .. Parameters ..
      real          ZERO, ONE, TWO, THREE, FOUR
      PARAMETER     (ZERO=0.0e0,ONE=1.0e0,TWO=2.0e0,THREE=3.0e0,
+               FOUR=4.0e0)
*      .. Scalar Arguments ..
      INTEGER       IFLAG, LDFJAC, N
*      .. Array Arguments ..
      real          FJAC(LDFJAC,N), FVEC(N), X(N)
*      .. Local Scalars ..
      INTEGER       J, K
*      .. Executable Statements ..
      IF (IFLAG.EQ.0) THEN
*
*         Insert print statements here when NPRINT is positive.
*
        RETURN
      ELSE
        IF (IFLAG.NE.2) THEN
          DO 20 K = 1, N
            FVEC(K) = (THREE-TWO*X(K))*X(K) + ONE
            IF (K.GT.1) FVEC(K) = FVEC(K) - X(K-1)

```

```

        IF (K.LT.N) FVEC(K) = FVEC(K) - TWO*X(K+1)
20      CONTINUE
      ELSE
        DO 60 K = 1, N
          DO 40 J = 1, N
            FJAC(K,J) = ZERO
40          CONTINUE
          FJAC(K,K) = THREE - FOUR*X(K)
          IF (K.GT.1) FJAC(K,K-1) = -ONE
          IF (K.LT.N) FJAC(K,K+1) = -TWO
60          CONTINUE
        END IF
      END IF
      RETURN
    END

```

9.2 Program Data

None.

9.3 Program Results

C05PCF Example Program Results

Final 2-norm of the residuals = 0.1193E-07

Number of function evaluations = 11

Number of Jacobian evaluations = 1

Final approximate solution

-0.5707	-0.6816	-0.7017
-0.7042	-0.7014	-0.6919
-0.6658	-0.5960	-0.4164
